Wiederholung: Worksheet 5 - 8

Wiederhole die Themen: Mittelwertsätze, Linearformen (Dualraum), Gruppenhomomorphismen, S_n, Signum, Determinante (Eindeutigkeit), Determinante (Existenz), Das Riemannintegral, Integrationstechniken sowie Jordan-Normalform!
Wir stellen Fragen im Testat!

Integrationstechniken: Partialbruchzerlegung

Aufgaben: 3

```
> restart;
```

Inhalt

Wir hatten bereits die partielle Integration und die Substitutionsregel behandelt. Hier soll die Rede von der Partialbruchzerlegung sein, die rationale Funktionen so vereinfacht, daß man sie möglicherweise leichter integrieren kann. Wir arbeiten über dem Körper $\mathbb{R}$ der reellen Zahlen.

MATH: Eine rationale Funktion ist der Quotient zweier Polynomfunktionen. Möglicherweise hat selbst im durchgekürzten Fall der Nenner noch Nullstellen, so daß die Funktion nicht mehr auf ganz $\mathbb{R}$ definiert ist.

BEISPIEL:

```
> f:=(x^2+1)/(x^3-1);
```

$$f := \frac{x^2+1}{x^3-1} \tag{2.1.1}$$

```
> simplify(f);
```

$$\frac{x^2+1}{x^3-1} \tag{2.1.2}$$

```
> plot(f,x=0..2);
```




```
> F:=int(f,x);
```

$$F := \frac{2}{3}\ln(x-1) + \frac{1}{6}\ln(x^2+x+1) - \frac{1}{3}\sqrt{3}\arctan\left(\frac{1}{3}(2x+1)\sqrt{3}\right) \qquad \textbf{(2.1.3)}$$

MAPLE kann offenbar besser integrieren, als wir es je von Hand können werden. Trotzdem wollen wir die Probe machen:

```
> diff(F,x);
  simplify(%);
  numer(%)/expand(denom(%));
```

$$\frac{2}{3(x-1)} + \frac{1}{6}\frac{2x+1}{x^2+x+1} - \frac{2}{3\left(1+\frac{1}{3}(2x+1)^2\right)}$$

$$\frac{x^2+1}{(x-1)(x^2+x+1)}$$

$$\frac{x^2+1}{x^3-1} \qquad \textbf{(2.1.4)}$$

MATH: Es stellt sich die Frage, ob eine Zerlegung von *f* als Summe, wie wir sie durch Integration und anschließende Differentiation erhalten haben, auch rein algebraisch erhalten werden kann. Das Zauberwort heißt **Partialbruchzerlegung**.

```
> convert(f,parfrac,x);
```

$$\frac{2}{3(x-1)} + \frac{1}{3}\frac{x-1}{x^2+x+1} \qquad \textbf{(2.1.5)}$$

Wir haben nur halbes Glück: Nur einer der beiden Terme läßt sich direkt integrieren. Aber auch der andere Term ist recht klar: Damit die Substitutionsregel angewandt werden kann, sorgen wir erst einmal dafür, daß im Zähler so die Ableitung des Nenners steht, dass der Rest ein konstantes

Polynom im Zähler hat. Wir erhalten also einen Summanden der Form $a \cdot \frac{p'}{p}$ und einen Summanden der Form $\frac{b}{p}$ mit $a, b \in \mathbb{R}$ und $p = x^2 + x + 1$.
Von diesen beiden Summanden kann man den ersten sehr leicht integrieren und für Terme in der Form des zweiten Summanden nutzt man man geeigneter Substitution die Identität $\arctan'(x) = \frac{1}{1+x^2}$ zur Integration.

ÜBUNG [01]:

Bestimme die (mit Hilfe von Maple, aber ohne **int**) die Stammfunktion von f auf folgende Art:
1) Führe die oben beschriebene Umformung und Analyse durch, um die gesuchte Zerlegung des Bruches zu erhalten, d.h. spalte den Term $\frac{1}{3} \frac{x-1}{x^2+x+1}$ weiter in zwei Summanden auf, wobei in einem der beiden Summanden der Zähler gerade die Ableitung des Nenners ist.
2) Integriere alle drei Summanden, die man aus $\frac{x^2+1}{x^3-1}$ erhalten hat.

Wir wiederholen kurz die Partialbruchzerlegung:

MATH: Es ist besser, die Partialbruchzerlegung für Quotienten von Polynomen statt für die induzierten Funktionen zu erklären:
1.) Schreibe $\frac{p}{q}$ für Polynome p, q als Summe eines Polynoms und eines echten Bruches.
(Dieser Schritt entfällt, wenn der Grad von p kleiner als der Grad von q ist.)

BEISPIEL:

```
> f:=(x+1)^3*(x^2+2)/(x^3+x+1);
```

$$f := \frac{(x+1)^3 (x^2+2)}{x^3+x+1} \tag{2.1.6}$$

```
> numer(f);denom(f);
```

$$(x+1)^3 (x^2+2)$$
$$x^3+x+1 \tag{2.1.7}$$

```
> quo(numer(f),denom(f),x);
```

$$x^2+3x+4 \tag{2.1.8}$$

ist der Polynomanteil und

```
> rem(numer(f),denom(f),x)/denom(f);
```

(2.1.9)

$$\frac{3\,x^2 - x - 2}{x^3 + x + 1} \tag{2.1.9}$$

ist der echte Bruchanteil.

MATH: 2.) Benutze den Euklidischen Algorithmus, um den verbleibenden echten Bruch zu kürzen.

3.) Faktorisiere den Nenner in irreduzible Faktoren.
Diese sind bekanntlich (über den reellen Zahlen) vom Grad eins oder zwei.

Hier ist eine gewisse Schwierigkeit: MAPLE kann diese Faktorisierung zunächst nur über $\mathbb{Q}$ durchführen.

```
> factor(x^3+x+1);
```

$$x^3 + x + 1 \tag{2.1.10}$$

Nach dem Zwischenwertsatz sollten wir wenigstens eine reelle Nullstelle haben:

```
> plot(x^3+x+1,x=-1..1);
```



Wir werden ihr den Namen α geben, und MAPLE mit diesem Namen weiterrechnen lassen:

```
> alias(alpha=RootOf(x^3+x+1));
```

$$\alpha \tag{2.1.11}$$

```
> factor(x^3+x+1,alpha);
```

$$-(-x+\alpha)\left(\alpha^2 + \alpha\,x + x^2 + 1\right) \tag{2.1.12}$$

Da $x^3 + x + 1$ monoton steigend ist, hat es keine weiteren reellen Nullstellen.

MATH: 4.) Ist der Nenner

$$P := \prod_{i=1}^{n} p_i^{a(i)}$$

mit verschiedenen irreduziblen Polynomen p_i, so zerlege den Bruch additiv in die Form

$$\sum_{i=1}^{n} \frac{z_i}{p_i^{a(i)}}$$

wieder mit Hilfe des Euklidischen Algorithmus, welcher auf die

$$Q_i := \frac{P}{p_i^{a(i)}}$$

anzuwenden ist, um Polynome R_i zu finden mit

$$\sum_{i=1}^{n} R_i Q_i = 1.$$

ÜBUNG [02]:

Führe Schritt 4.) für den vorliegenden Fall

$$\frac{-2+3\cdot x^2 - x}{(x-\alpha)\cdot\left(x^2+\alpha\, x+1+\alpha^2\right)}$$

durch. (Hinweis: Benutze gcdex)

```
> (-2+3*x^2-x)/((x-alpha)*(x^2+alpha*x+1+alpha^2));
```

$$\frac{3\,x^2 - x - 2}{(x-\alpha)\,\left(\alpha^2+\alpha\, x+x^2+1\right)} \tag{2.1.13}$$

MATH: 5.) Nach Ausführung von Schritt 1) kann man ohne Einschränkung davon ausgehen, dass jeder Summand

$$\frac{z}{p^a} := \frac{z_i}{p_i^{a(i)}}$$

$Grad(z) < Grad(p^a)$ erfüllt. Schreibe diesen Summanden als

$$\frac{z}{p^a} = \frac{A_1}{p} + \frac{A_2}{p^2} + \ldots + \frac{A_a}{p^a}$$

im Falle $p = x - w \in \mathbb{R}[x]$ vom Grad 1 mit $A_j \in \mathbb{R}$

und

$$\frac{z}{p^a} = \frac{B_1 + x\cdot C_1}{p} + \frac{B_2 + x\cdot C_2}{p^2} + \,..\, + \frac{B_a + x\cdot C_a}{p^a}$$

im Falle $p = x^2 + v\cdot x + u \in \mathbb{R}[x]$ vom Grad 2 mit $B_j, C_j \in \mathbb{R}$.

Unsere Funktion f war so einfach gewählt, dass dieser Schritt nicht notwendg wird.

Wir führen dies an an einem anderen Beispiel durch, so dass klar wird, wie dies allgemein geht:

```
> s:=add(x^i,i=0..5)/(x-2)^6;
```

$$s := \frac{x^5+x^4+x^3+x^2+x+1}{(x-2)^6} \quad (2.1.14)$$

```
> expand(subs(x=y+2,numer(s)));
```

$$y^5+11\,y^4+49\,y^3+111\,y^2+129\,y+63 \quad (2.1.15)$$

```
> subs(y=``(x-2),%);
```

$$(x-2)^5+11\,(x-2)^4+49\,(x-2)^3+111\,(x-2)^2+129\,(x-2)+63 \quad (2.1.16)$$

Damit ist klar, wie es sich im linearen Fall verhält. Man hätte auch schneller den Taylorbefehl benutzen können:

```
> taylor(numer(s),x=2);
```

$$63+129\,(x-2)+111\,(x-2)^2+49\,(x-2)^3+11\,(x-2)^4+(x-2)^5 \quad (2.1.17)$$

Beachte: Die obigen Koeffizienten 63, 129, 111 etc. erhält man als den Reste: 63 als Rest von *numer(s)* modulo *(x-2)*. Man subtrahiert 63, dividiert durch *(x-2)* und erhält 129 als den Rest *mod (x-2)* etc.

freiwillige ÜBUNG:

Führe die oben beschriebene Zerlegung von Schritt 5.) durch für

```
> t:=add(x^i,i=0..5)/(x^2+x+2)^4;
```

$$t := \frac{x^5+x^4+x^3+x^2+x+1}{(x^2+x+2)^4} \quad (2.1.18)$$

Wir haben in diesem Abschnitt bereits (fast) alles gemacht um $f = \frac{(x+1)^3 \cdot (x^2+2)}{x^3+x+1}$ zu integrieren.

ÜBUNG [03]:

1) Fasse zusammen, welche Zerlegungen von $f = \frac{(x+1)^3 \cdot (x^2+2)}{x^3+x+1}$ in Summanden bereits gemacht wurden, um die Integration vorzubereiten.
2) Zerlege f weiter in eine Summe von Termen, bis du alle Summanden (von Hand) integrieren könntest (zum Beispiel mit den in Aufgabe 1 verwendeten Integrationsverfahren).
3) Integriere alle Summanden. (Es kann zwar der **int**-Befehl benutzt werden, aber wir erwarten, dass du genau beschreiben kannst, welche Integrationsregeln aus welchem Grund angewendet werden.)

```
> f ;
```

(2.1.19)

$$\frac{(x+1)^3\,(x^2+2)}{x^3+x+1} \qquad \textbf{(2.1.19)}$$

MATH: Wir wollen uns noch davon überzeugen, das man (oder MAPLE) auch s und t leicht integrieren kann:

```
> int(s,x);
```

$$-\frac{49}{2\,(x-2)^2}-\frac{37}{(x-2)^3}-\frac{63}{5\,(x-2)^5}-\frac{11}{x-2}+\ln(x-2)-\frac{129}{4\,(x-2)^4} \qquad \textbf{(2.1.20)}$$

```
> int(t,x);
```

$$\frac{1}{7}\,\frac{-3\,x-5}{x^2+x+2}-\frac{2}{343}\sqrt{7}\arctan\left(\frac{1}{7}\,(2\,x+1)\sqrt{7}\right)+\frac{1}{21}\,\frac{-7\,x-7}{(x^2+x+2)^3}$$
$$-\frac{5}{42}\,\frac{2\,x+1}{(x^2+x+2)^2}+\frac{10}{49}\,\frac{2\,x+1}{x^2+x+2}+\frac{1}{14}\,\frac{10\,x+12}{(x^2+x+2)^2} \qquad \textbf{(2.1.21)}$$

Bei t kommt wieder der anfangs behandelte Trick mit der quadratischen Ergänzung zum Tragen, allerdings in iterierter Form.

DENKANSTOSS: Man könnte komplex vorgehen und hätte dann eine einfachere Partialbruchzerlegung. Man würde sich aber komplexe Logarithmen einhandeln.

DENKANSTOSS: Zur Integration von rationalen Funktionen fehlt noch die Stammfunktion von $\frac{1}{(x^2+1)^n}$, da wir alle anderen Summanden der Partialbruchzerlegung mit den obigen Methoden im Griff haben. Betrachte dazu:

```
> normal(diff(x/(x^2+1)^2,x));
  normal(diff(x/(x^2+1)^3,x));
  normal(diff(x/(x^2+1)^4,x));
```

$$-\frac{3\,x^2-1}{(x^2+1)^3}$$
$$-\frac{5\,x^2-1}{(x^2+1)^4}$$
$$-\frac{7\,x^2-1}{(x^2+1)^5} \qquad \textbf{(2.1.22)}$$

um eine Inspiration für eine Formel zu erhalten und nutze dann Substitution, um andere irreduzible Faktoren vom Grad 2 auf diesen zurück zu führen.

Integrale als Konstruktion neuer Funktionen:

Logarithmus

Aufbauend auf den Abschnitten: "Definition des Riemannintegrals", "Der Fundamentalsatz der Differential- und Integralrechnung"

Aufgaben: 1

```
> restart;
```

Inhalt

Man kann den Fundamentalsatz der Integral- und Differentialrechnung auch so verstehen, dass eine neue Funktion durch eine gegebene konstruiert wird (eben als Integralfunktion). Man kann die Eigenschaften des Integrals benutzen, um Eigenschaften der neuen Funktion herzuleiten.

Wir wollen jetzt die Stammfunktion von $x \to \frac{1}{x}$ auf dem offenen Intervall $(0, \infty)$ diskutieren.

```
> L:=unapply(int(1/t,t=1..x),x);
```

Warning, unable to determine if 0 is between 1 and x; try to use assumptions or use the AllSolutions option

$$L := x \to \int_1^x \frac{1}{t}\,\mathrm{d}t \tag{3.1.1}$$

```
> L(x*y);
```

$$\int_1^{xy} \frac{1}{t}\,\mathrm{d}t \tag{3.1.2}$$

Dies sieht fast wie die rechte Seite der Substitutionsregel aus. In der Tat, $L(x \cdot y)$ ist offenbar gleich

```
> L(x)+int(1/(t),t=x..x*y);
```

Warning, unable to determine if 0 is between x and x*y; try to use assumptions or use the AllSolutions option

$$\int_1^{x} \frac{1}{t}\,\mathrm{d}t + \int_x^{xy} \frac{1}{t}\,\mathrm{d}t \tag{3.1.3}$$

MAPLE kann verifizieren, daß das zweite Integral gleich $L(y)$ ist:

```
> simplify(int(1/(u),u=x..x*y)-L(y)) assuming x>0 assuming
  y>0;
```

$$0 \tag{3.1.4}$$

DENKANSTOSS: Kannst du das auch mit Hilfe der Substitutionsregel verifizieren? Was ist die geometrische Bedeutung dieser erstaunlichen Identität?

Wir halten fest:

MATH: Für $x, y > 0$ gilt die Funktionalgleichung

$$L(x \cdot y) = L(x) + L(y).$$

DENKANSTOSS: Benutze die Definition von L, um

1) $L(x) > 0$ für $x > 1$,
1') $L(x) < 0$ für $x < 1$,
2) L monoton steigend

nachzuweisen. Benutze die Funktionalgleichung, um

3) $\lim_{x\to\infty} L(x) = \infty$,
3') $\lim_{x\to 0+} L(x) = -\infty$

nachzuweisen.

ÜBUNG [04]:

1) Folgere aus der Homomorphieeigenschaft der Funktion *L* von oben (Multiplikation im Definitionsbereich überträgt sich zu Addidtion bei den Werten), dass sie auf der postiven reellen Achse sowohl nach oben als auch nach unten unbeschränkt ist (Beweis!).
2) Zeige, dass $L: (0,\infty) \to \mathbb{R}$ bijektiv ist.
3) Finde danach die Ableitung der Umkehrfunktion von L.
4) Zeige, dass eine Funktion $f: \mathbb{R} \to \mathbb{R}$, die $f' = f$ erfüllt, ein Vielfaches der Exponentialfuntion ist.
Hinweis: Leite die Funktion $g := f \cdot \exp(-x)$ ab.
5) Zeige, dass die Umkehrfunktion der Funktion *L* die Expontentialfunktion ist. (Insbesondere gilt damit $L = \ln$).
6) Bestimme die Stammfunktion der Tangensfunktion. Der Tangens ist definiert als:

```
> convert(tan(x),sincos);
```

$$\frac{\sin(x)}{\cos(x)} \qquad (3.1.5)$$

MAPLE: Nachdem wir MAPLE mitgeteilt haben, daß x reell positiv ist, weiß es, daß *L* und $\ln$ dasselbe sind. Wir haben dieses Wissen nicht benutzt.

```
> L(x) assuming x>0;
```

$$\ln(x) \qquad (3.1.6)$$

MATH: Die Funktionalgleichung sagt uns, daß der Logarithmus einen Homomorphismus von der multiplikativen Gruppe der positiven reellen Zahlen in die additive Gruppe (=,+) aller reellen Zahlen darstellt.

$$L: \left(\mathbb{R}_{>0}, \cdot\right) \to (\mathbb{R}, +)$$

Dies bedeutet anschaulich, daß multiplikative Phänomene auf einer additiven Skala dargestellt werden.

So heißt z. B.

```
> diff(L(f(x)),x);
```

$$\frac{\frac{d}{dx} f(x)}{f(x)} \qquad (3.1.7)$$

für positives *f* die logarithmische Ableitung, die uns multiplikatives Wachstum auf einer additiven Skala repräsentiert:

```
> dsolve(diff(L(f(x)),x)=g(x),f(x));
```

$$f(x) = \_C1\, e^{\int g(x)\, dx} \qquad (3.1.8)$$

Beachte insbesondere den Fall *g(x)=1*.

Uneigentliche Integrale

Aufgaben: 3

```
> restart;
```

Uneigentliche Integrale

Das Integral ist bereits ein Grenzwert. Kann man die obere Grenze gegen unendlich oder die untere gegen -unendlich laufen lassen oder eines von beiden gegen eine Unendlichkeitsstelle (Polstelle) der Funktion, so kann man wieder Grenzwertbetrachtungen anstellen. Diese sind für sich interessant, lassen sich aber auch auf Konvergenzbetrachtungen von Reihen anwenden, weil der Fundamentalsatz oft die explizite Bestimmung der Integralfunktionen ermöglicht.

MATH: Es gibt diverse Situationen, wo man über uneigentliche Riemannintegrierbarkeit spricht. Wir wollen erstmal nur eine hervorheben:

$$f\colon [a, \infty) \to \mathbb{R}$$

heißt uneigentlich Riemannintegrierbar, falls die Einschränkung von *f* auf jedes Intervall [*a,b*] mit $b > a$ Riemannintegrierbar ist und

$$\lim_{b \to \infty} \int_a^b f(x)\, dx$$

existiert (als reelle Zahl).

DENKANSTOSS: Man ziehe eines der Vergleichskriterien aus der Vorlesung heran, um für positive Funktionen *f* die Konvergenz von $f(x) \to 0$ für $x \to \infty$ gegen unendlich als notwendige Bedingung für uneigentliche Integrierbarkeit nachzuweisen.

MAPLE kann uneigentliche Riemannintegrierbarkeit oftmals leicht feststellen, z. B. weil es die Stammfunktion formal ausrechnen kann.

Wir haben auch schon ein paar Integrationstechniken kennengelernt und wollen diese hier anwenden.

ÜBUNG [05]:

Bei dieser Aufgabe soll ausdrücklich *nicht* der **Int**-Befehl von Maple benutzt werden. Der **limit**-Befehl darf aber verwendet werden.

1) Sei $t \in \left[0, \frac{\pi}{2}\right]$. Zeige: $\cos(t) = \frac{1}{\sqrt{\tan(t)^2+1}}$.

2) Folgere: $\cos(\arctan(x)) = \frac{1}{\sqrt{x^2+1}}$ für $x > 0$.

Welche der folgenden Funktionen ist uneigentlich Riemannintegrierbar (für $x \to \infty$)?

3) $\arctan(x)$ (Hinweis: Integration der Umkehrfunktion)

4) $\frac{\pi}{2} - \arctan(x)$

5) $\frac{1}{x^2+1}$

6) $\frac{1}{x^2-1}$ (Hinweis: Partialbruchzerlegung)

Integralvergleichskriterium für Reihen

MATH: (Integralvergleichskriterium für Reihen): Sei

$$f : [1, \infty) \to \mathbb{R}$$

monoton fallend. Dann sind

$$\sum_{k=1}^{\infty} f(k)$$

und

$$\int_1^{\infty} f(x)\,dx$$

entweder beide divergent oder beide konvergent.

Die Anwendung ist meistens so, dass man die Konvergenz des Integrals entscheiden kann und dann auf die Reihe schließt. Der Grenzwert der Reihe ist im Falle der Konvergenz dann aber immer noch nicht bestimmt.

DENKANSTOSS: Vergleiche, durch Transformation der Integrale, auch mit dem Verdichtungslemma: Unter den obigen Voraussetzungen sind

$$\sum_{k=1}^{\infty} f(k)$$

und

$$\sum_{k=1}^{\infty} 2^k f(2^k)$$

entweder beide konvergent oder beide divergent.

ÜBUNG [06]:

1) Ist $\frac{\ln(n)}{n^2}$ monoton fallend?

2) Benutze das obige Vergleichskriterium, um die Konvergenz von

$$\sum_{n=1}^{\infty} \frac{\ln(n)}{n^2}$$

zu entscheiden.

Eulersche Integrale

Hier geht es um die Eulersche Betafunktion und ihren Zusammenhang mit der Gammafunktion, welche die Fakultätsfunktion interpoliert. Die Betafunktion ist verwandt mit dem Kehrwert der Binomialkoeffizienten:

```
> Beta(p,q)=Int(x^(p-1)*(1-x)^(q-1),x=0..1);
```

$$\mathrm{B}(p, q) = \int_0^1 x^{p-1} (1-x)^{q-1} \, \mathrm{d}x \qquad \textbf{(4.3.1)}$$

```
> Beta(1,q);
```

$$\frac{1}{q} \qquad \textbf{(4.3.2)}$$

(Das griechische "grosse β" sieht aus wie ein B.)

ÜBUNG [07]:

1) Für welche (reellen) Werte von p und q ist der obige Ausdruck $\mathrm{B}(p, q)$ ein uneigentliches Integral?
2) Wann ist es divergent?
Hinweis: Unterteile das Intervall $[0, 1]$ und schätze jeweils einen Faktor des Integraden ab.

MATH: Die Gammafunktion

$$\Gamma : x \mapsto \int_0^{\infty} t^{x-1} \mathrm{e}^{-t} \, \mathrm{d}t$$

```
> int(t^(x-1)*exp(-t),t=0..infinity);
```

$$\Gamma(x) \qquad \textbf{(4.3.3)}$$

interpoliert die Funktion

$$\mathbb{N} \to \mathbb{N} : n \to (n-1)!\ ,$$

wie man durch iterierte partielle Integration sieht.

```
> GAMMA(1),0!;
  GAMMA(2),1!;
  GAMMA(3),2!;
  GAMMA(4),3!;
  GAMMA(5),4!;
```

$$1,\ 1$$

$$1,\ 1$$

$$2,\ 2$$

$$6,\ 6$$

$$24,\ 24 \tag{4.3.4}$$

Man kann B mit hilfe von Γ darstellen

```
> int(x^(p-1)*(1-x)^(q-1),x=0..1);
```

$$\frac{\Gamma(q)\,\Gamma(p)}{\Gamma(p+q)} \tag{4.3.5}$$

und erkennt die Ähnlichkeit zum Kehrwert des Binomialkoeffizienten.